
Subject: Re: Colorbar

Posted by [David Fanning](#) on Thu, 13 Jan 2011 21:40:32 GMT

[View Forum Message](#) <> [Reply to Message](#)

kisCA writes:

>
> Ok thanks, your commands are quite amazing!
>
> To explain me better maybe an example will help:
> The original range of my data is [-1.06371,0.907327].
> I would like to draw fill contour from -0.8 to 0.8 with an increment
> of 0.1 but keeping the color scale between the original range...
> Is it better this way?

I would do this:

```
data = Scale_Vector(Dist(41,41), -1.06371 ,0.907327)
levels = [Min(data), Scale_Vector(findgen(17), -0.8, 0.8)]
pos = [0.1, 0.1, 0.9, 0.75]
LoadCT, 33, NColors=16, Bottom=2
TVLCT, FSC_Color('light gray', /Triple), 1
TVLCT, FSC_Color('dark gray', /Triple), 18
FSC_Display, 800, 500
FSC_Contour, data, Levels=levels, /Fill, $
    C_Colors=Indgen(20)+1, Position=pos
FSC_Contour, data, Levels=levels, /Overplot, Label=2
labels = ['<-0.8', String(Scale_Vector(findgen(16), -0.8, 0.7),$
    Format='(F0.1)'), '>0.8']
DCBar, NColors=18, Bottom=1, Labels=labels, charsize=0.75
```

This will give you a light grey color for values less than -0.8 and a dark gray color for values greater than 0.8. Values between -0.8 and 0.8 will be in increments of 0.1 and in normal colors.

Cheers,

David

--

David Fanning, Ph.D.

Fanning Software Consulting, Inc.

Coyote's Guide to IDL Programming: <http://www.idlcoyote.com/>

Sepore ma de ni thui. ("Perhaps thou speakest truth.")
